
//------------------------------------------------------------------------
Params
	Bool lotskg(True); //True为Fund自动计算下单手数,False为lots_data1和lots_data1手动设置下单数量
	Numeric lots_data1(1);//自定义data0品种手数
	Numeric lots_data2(1);//自定义data1品种手数
	Numeric Fund(20000);//保证金自动换算
	Numeric N(60); //周期参数1
	Numeric M(7); //周期参数2
	Numeric X(5); //周期参数3
		//Bar更新事件函数，参数indexs表示变化的数据源图层ID248844320数组
	Numeric Stopout(3); //出场参数
Vars

	Series<Numeric> ma1;
	Series<Numeric> star_pirce;
	Series<Numeric> D_jcprice;
	Series<Numeric> K_jcprice;
	Series<Numeric> stopprice;
	Series<Numeric> copen;
	Series<Numeric> chigh;
	Series<Numeric> clow;
	Series<Numeric> Stopof;
	Series<Numeric> arass;
	Series<Numeric> larass;
	
	Plot plt1;
	Plot plt2;
	Series<Numeric> jc;
	Series<Numeric> c_jb;
	Series<Numeric> h_jb;
	Series<Numeric> l_jb;
	Series<Numeric> o_jb;
	Series<Bool> cond_dk;
	Series<Bool> cond_kk;
	Numeric Lots(0);
	Series<Numeric> HD;
	Series<Numeric> LD;
	Series<Bool> DK_up;
	Series<Bool> KK_dn;
	Series<Bool> condD;
	Series<Bool> condK;
	Series<Numeric> Curbar_H;
	Series<Numeric> Curbar_L;
	Series<Numeric> Highup;
	Series<Numeric> Lowdown;
	Series<Numeric> BD1;
	Series<Numeric> BD2;
	Series<Numeric> data1_kg;
	Series<Numeric> data2_kg;
	
Defs

	//基础手数计算
	Integer init_lots()
	{

		if(lotskg==true)
		{
			Lots=Max(1,IntPart(Fund/(O/Rollover *ContractUnit*BigPointValue*0.1))); //计算开仓手数，按照除权后真实价格计算
		}Else
		{
			Lots=lots_data1;
		}
		Return 1;
	}
	//画线
	Integer plot_mod()
	{
						
		plt1.line("BD1",BD1);
		plt1.line("BD2",BD2);
		plt2.line("c_jb",c_jb); //价差线
		plt2.line("arass",arass); //出场线
		Return 1;
	}
	//核心代码开仓逻辑，仅需要在这里编写开仓逻辑即可
	Integer buyorsell()
	{
			// 定义一个名为 KG 的数值变量，初始值为 0
		    Numeric KG=0;
		    // 定义多个数值变量，分别表示前一根K线和当前K线的开盘价、收盘价、最高价和最低价
			Numeric c1=Data0.Close;
			Numeric c2=Data1.Close;
			Numeric o1=Data0.open;
			Numeric o2=Data1.open;
			Numeric ch1=Data0.high;
			Numeric ch2=Data1.high;
			Numeric cl1=Data0.low;
			Numeric cl2=Data1.low;
			// 计算当前价差
			c_jb=c1/c2;
			copen = (o1/o2);//最新价差
			chigh = max(max(ch1/ch2,o1/o2),c1/c2);//最新价差
			clow = min(min(cl1/cl2,o1/o2),c1/c2);//最新价差	
			
			/* 指标逻辑解释：
				计算价格波动范围：

				通过在指定周期内计算最高价和最低价的差值，确定价格的波动范围。这些基础计算用于后续的归一化和百分比计算。
				归一化和百分比计算：

				计算当前价格相对于过去 N 个周期的最高价和最低价的位置。通过这些计算，可以得到当前价格在最近波动范围内的相对位置（如百分比表示）。
				加权价格计算：

				通过对当前价格、最高价、和最低价的加权平均值计算一个平衡价格。这有助于进一步平滑价格数据，减少波动影响。
				移动平均线的计算：

				使用简单移动平均（SMA）和加权移动平均（XAverage）来平滑计算得出的指标。通过这些移动平均线，可以识别出价格趋势的变化。
				趋势反转的识别（交叉点）：

				代码通过比较两个移动平均线（BD1 和 BD2）之间的交叉点来识别趋势反转的信号：
				金叉（DK_up）：当较短期的移动平均线（BD1）向上穿过较长期的移动平均线（BD2）时，通常被视为买入信号。
				死叉（KK_dn）：当较短期的移动平均线（BD1）向下穿过较长期的移动平均线（BD2）时，通常被视为卖出信号。			
			*/
			
			// 定义变量 VAR1 为 N 个周期内的最高价与最低价的差值，表示价格波动范围
			Numeric VAR1 = (Highest(chigh, N) - Lowest(clow, N));
			// 定义变量 VAR2 为 N 个周期内的最高价与当前价格（c_jb）的差值，表示当前价格离最高价的距离
			Numeric VAR2 = (Highest(chigh, N) - c_jb);
			// 定义变量 VAR3 为当前价格（c_jb）与 N 个周期内最低价的差值，表示当前价格离最低价的距离
			Numeric VAR3 = (c_jb - Lowest(clow, N));
			// 定义变量 VAR4 为 VAR2 相对于 VAR1 的百分比，并减去 70，用于衡量当前价格在最近 N 个周期内的相对位置
			Numeric VAR4 = VAR2 / VAR1 * 100 - 70;
			// 定义变量 VAR5 为 VAR3 相对于 VAR1 的百分比，用于表示当前价格在 N 个周期内的相对百分比位置
			Numeric VAR5 = (c_jb - Lowest(clow, N)) / (Highest(chigh, N) - Lowest(clow, N)) * 100;
			// 定义变量 VAR6 为加权价格，使用当前价格（c_jb）、最高价（chigh）和最低价（clow）的加权平均值
			Numeric VAR6 = (2 * c_jb + chigh + clow) / 4;
			// 定义变量 VAR7 为 VAR3 相对于 VAR1 的百分比的 3 日简单移动平均线，用于平滑波动
			Numeric VAR7 = SMA((VAR3 / VAR1 * 100), 3, 1);
			// 定义变量 VAR8 为 M 个周期内的最低价
			Numeric VAR8 = Lowest(clow, M);
			// 定义变量 VAR9 为 VAR7 的 3 日移动平均值与 VAR4 的 9 日移动平均值之间的差值，用于识别价格动能的变化
			Numeric VAR9 = SMA(VAR7, 3, 1) - SMA(VAR4, 9, 1);
			// 定义变量 VARA 为 M 个周期内的最高价
			Numeric VARA = Highest(chigh, M);
			// 定义变量 BD1 为加权价格相对于最高价和最低价的位置的归一化值，并计算其 M 日的加权移动平均线
			BD1 = XAverage((VAR6 - VAR8) / (VARA - VAR8) * 100, M);
			// 定义变量 BD2 为 BD1 的 X 日加权移动平均线
			BD2 = XAverage(BD1, X);
			// 定义变量 ma1 为当前价格（c_jb）的 X 日加权移动平均线
			ma1 = XAverage(c_jb, X);
			// 定义 DK_up 为 BD1 上穿 BD2 的信号，用于识别金叉（买入信号）
			DK_up = CrossOver(BD1, BD2);  // 金叉
			// 定义 KK_dn 为 BD1 下穿 BD2 的信号，用于识别死叉（卖出信号）
			KK_dn = CrossUnder(BD1, BD2);  // 死叉

		
			If(DK_up)//获得金叉时的BAR线索引号
			{
				Curbar_H = CurrentBar;			
			}
			If(KK_dn)//获得死叉时的BAR线索引号
			{
				Curbar_L = CurrentBar;			
			}
			Highup  = Highest(chigh, Max(Curbar_H - Curbar_L, M));//区间的周期数最小阀值限制,最小为M
			Lowdown = Lowest (clow, Max(Curbar_L - Curbar_H, M));//区间的周期数最小阀值限制,最小为M
		
			If(DK_up[1])//前一根bar线金叉时保存highup到HD
			{				
				HD = Highup;			
			}
			If(KK_dn[1])//前一根bar线死叉时保存lowdown到LD
			{			
				LD = Lowdown;			
			}
			
			cond_dk=BD1 > BD2 and chigh >= HD[1] and c_jb[1] > ma1[1];
			cond_kk=BD1 < BD2 and clow <= LD[1] and c_jb[1] < ma1[1];

			KG=IIF(cond_dk,1,IIF(cond_kk,-1,0));
			
			Return KG;
	} 
	
	//价差线多头跟随出场
	Integer pingduo_long()
	{
		Numeric Agg=((D_jcprice-chigh)/chigh); //计算当前价位与最高价位的比例
		bool cond_Agg=Agg>0; //判断当前价位是否高于最高价位
		if(cond_Agg) //如果当前价位高于最高价位
		{
			Stopof=Stopout*(1-Agg*10); //根据比例设置止损/止盈参数
		}Else
		{
			Stopof=Stopof; //如果当前价位不高于最高价位，则保持原有止损/止盈参数不变
		}
		D_jcprice=max(D_jcprice,c_jb); //更新止损/止盈参考价位
		stopprice=D_jcprice-(Stopout/100)*c_jb[1]; //根据当前价位和止损/止盈参数计算止损价位
		if(BarsSinceEntry==0 ) //如果当前是第一根K线，则将stopprice设为止损价位
		{
			arass=stopprice;// 将 arass 变量的值更新为止损价格
		}Else
		{
			arass=max(D_jcprice-(Stopof/100)*c_jb[1],arass); //否则根据当前价位、止损/止盈参数和原有arass计算新的arass
		}
		Return 1;
	}
	//价差线空头跟随出场
	Integer pingkong_short()
	{
		Numeric Agg = ((K_jcprice - clow) / clow);//计算当前价位与最高价位的比例
		bool cond_Agg = Agg < 0; //判断当前价位是否低于最低价位
		if (cond_Agg) //如果当前价位低于最低价位
		{
			Stopof = Stopout * (1 + Agg * 10);//根据比例设置止损/止盈参数
		}
		Else 
		{
			Stopof = Stopof;//如果当前价位不低于最低价位，则保持原有止损/止盈参数不变
		}
		K_jcprice = min(K_jcprice, c_jb); //更新止损/止盈参考价位
		stopprice = K_jcprice + (Stopout / 100) * c_jb[1];//根据当前价位和止损/止盈参数计算止损价位
		if (BarsSinceEntry == 0) //如果当前是第一根K线，则将stopprice设为止损价位
		{
			arass = stopprice;// 将 arass 变量的值更新为止损价格
		} Else 
		{
			arass = min(K_jcprice + (Stopof / 100) * c_jb[1], arass);//否则根据当前价位、止损/止盈参数和原有arass计算新的arass
		}
		Return 1;
	}
	
	
Events
	OnInit()
	{
		//与数据源有关
		Range[0:DataCount-1]
		{
			//=========数据源相关设置==============
			AddDataFlag(Enum_Data_RolloverBackWard());	//设置后复权

			AddDataFlag(Enum_Data_RolloverRealPrice());	//设置映射真实价格

			AddDataFlag(Enum_Data_AutoSwapPosition());	//设置自动换仓


			//AddDataFlag(Enum_Data_IgnoreSwapSignalCalc());	//设置忽略换仓信号计算
			plt2.figure(0);
			plt1.figure(0);
		}
	}
    onBar(ArrayRef<Integer> indexs)
    {    
		Range[0:0]
        {  
        
			init_lots(); //基础的手数计算
			//核心逻辑计算返回 1，-1，0
			data1_kg=buyorsell();
			//开多模块
			if(data1_kg[1]>0) //
			{
				if(MarketPosition==0) //当前市场仓位为0，则执行买入操作
				{
					Buy(Lots,open); //开多
					D_jcprice=c_jb; //设置当前价位为止损/止盈参考价位
					arass=0; //将 arass 变量清零
				}   
			} 
			
			if(MarketPosition>0 and D_jcprice>0)
			{
				//价差线多头跟随出场
				pingduo_long();
				if(c_jb[1]<arass[1] and BarsSinceEntry>0) //平多条件
				{
					sell(0,open); //平多
					D_jcprice=0; //重置开仓价格
				}
			}

			// 开空模块
			if (data1_kg[1]< 0) 
			{
				if (MarketPosition == 0) // 如果当前没有持仓
				{
					SellShort(Lots, open);// 开空
					K_jcprice = c_jb;// 将当前收盘价赋值给 K_jcprice 变量
					arass = 0;// 将 arass 变量清零
				}		
			}
			
			if (MarketPosition < 0 and K_jcprice > 0) 
			{
				//价差线空头跟随出场
				pingkong_short();
				if (c_jb[1] > arass[1] and BarsSinceEntry > 0) // 如果当前价格高于空头出场线且持仓天数大于 0 平仓
				{
					buytocover(0, open);// 平空
					K_jcprice = 0;//重置价格
				}
			}
			//画线
			plot_mod();
		}

		Range[1:1]
		{
			/*注意：模块是与Range[0:0]一致的，只是下单方向相反------------------------------------------------------*/
			
			init_lots(); //基础的手数计算
			//核心逻辑计算返回 1，-1，0
			data2_kg=buyorsell();
			//开空模块
			if(data2_kg[1]>0)   // 判断是否大于0
			{
				if(MarketPosition==0)   // 如果当前持仓为0
				{
					SellShort(Lots,open);   // 卖空开仓
					D_jcprice=c_jb;   // 记录开仓价
					arass=0;   // 重置止损跟踪价
				}
			} 
			if(MarketPosition<0 and D_jcprice>0)
			{
				//价差线多头跟随出场
				pingduo_long();
			
				if(c_jb[1]<arass[1] and BarsSinceEntry>0)  // 如果当前价格低于止损跟踪价平仓
				{
					buytocover(0,open);   // 平仓
					D_jcprice=0;   // 重置开仓价
				}
			}
			//开多模块
			if (data2_kg[1]<0)  //如果自定义函数小于0（下跌）
			{
				if(MarketPosition==0) //如果没有持仓
				{
					Buy(Lots,open); //开多仓
					K_jcprice=c_jb; //记录买入价格
					arass=0; //清空交易的止损价
				}			
			}
			
			if (MarketPosition > 0 and K_jcprice > 0) 
			{
				//价差线空头跟随出场
				pingkong_short();
				if(c_jb[1]>arass[1] and BarsSinceEntry>0) //如果当前价格超过了止损价平仓
				{
					Sell(0,open); //平多仓
					K_jcprice=0; //重置开仓价
				}
			}
		}
    
    }

